A GAMS TUTORIAL

WHAT IS GAMS ?

- General Algebraic Modeling System
- Modeling linear, nonlinear and mixed integer optimization problems
- Useful with large, complex problems

A GAMS Example
TRANSPORTATION EXAMPLE

- A toy problem!…
- 2 supply plants, 3 markets, and 1 commodity.
- Given: unit costs of shipping.
- How much to ship to minimize total transportation cost

A GAMS Example
TRANSPORTATION EXAMPLE



Minimize: Transportation cost

Subject to: Demand satisfaction and supply constraints

A GAMS Example
TRANSPORTATION EXAMPLE

Distances				
	Markets			
Plants	**New York**	**Chicago**	**Topeka**	**Supply**
Seattle	2.5	1.7	1.8	350
San Diego	2.5	1.8	1.4	600
Demand	325	300	275	

Shipping costs are assumed to be $90 per case per kMile.

Transportation Example: **ALGEBRAIC REPRESENTATION**

Indices (or sets):

i = plants

j = markets

Given Data (or parameters):

a_i = supply of commodity of plant i (in cases)

b_j = demand for commodity at market j (cases)

c_{ij} = cost per unit shipment between plant i and market j ($/case)

Transportation Example: ALGEBRAIC REPRESENTATION

Decision Variables:

x_{ij} = amount to ship from plant i to market j (cases),

where $x_{ij} \geq 0$, for all i, j

Constraints:

Observe supply limit at plant i: $\sum_j x_{ij} \leq a_i$, *for all i* (cases)

Satisfy demand at market j: $\sum_i x_{ij} \geq b_j$, *for all j* (cases)

Objective Function:

Minimize $\sum_i \sum_j c\, x_{ij}$ ($K)

Transportation Example: **ALGEBRAIC REPRESENTATION**

- All the entities of the model are identified (and grouped) by type.
- the ordering of entities is chosen so that no symbol is referred to before it is defined.

gamside: C:\WINDOWS\gamsdir\project.gpr - [C:\WINDOWS\gamsdir\trnsport.gms]

File Edit Search Windows Help

trnsport.gms | trnsport.lst

```
Sets
       i   canning plants    / seattle, san-diego /
       j   markets           / new-york, chicago, topeka / ;

Parameters

       a(i)  capacity of plant i in cases
         /    seattle     350
              san-diego   600  /

       b(j)  demand at market j in cases
         /    new-york    325
              chicago     300
              topeka      275  / ;

Table d(i,j)  distance in thousands of miles
                  new-york       chicago      topeka
    seattle          2.5           1.7          1.8
    san-diego        2.5           1.8          1.4  ;

Scalar f  freight in dollars per case per thousand miles  /90/ ;

Parameter c(i,j)  transport cost in thousands of dollars per case ;

          c(i,j) = f * d(i,j) / 1000 ;
```

40: 37 | Insert

gamside: C:\WINDOWS\gamsdir\project.gpr - [C:\WINDOWS\gamsdir\trnsport.gms]

File Edit Search Windows Help

trnsport.gms | trnsport.lst

```
Variables
     x(i,j)  shipment quantities in cases
     z       total transportation costs in thousands of dollars ;

Positive Variable x ;

Equations
     cost          define objective function
     supply(i)     observe supply limit at plant i
     demand(j)     satisfy demand at market j ;

cost ..        z  =e=  sum((i,j), c(i,j)*x(i,j)) ;

supply(i) ..   sum(j, x(i,j))  =l=  a(i) ;

demand(j) ..   sum(i, x(i,j))  =g=  b(j) ;

Model transport /all/ ;

Solve transport using lp minimizing z ;

Display x.l, x.m ;
```

40: 37 Insert

Transportation Example:
THE GAMS MODEL (Sets)

```
Sets
      i    canning plants    / seattle, san-diego /
      j    markets           / new-york, chicago, topeka / ;
```

- Declare and name the sets
- Assign their members between slashes / … /
- Multiword names are not allowed “New York”, use hyphens “New-York”
- Terminate every statement with a semicolon (;).
- compiler does not distinguish between upper- and lowercase letters.

Transportation Example:
THE GAMS MODEL (Sets)

We can put sets into separate statements:

```
Set i canning plants / Seattle, San-Diego / ;
Set j markets / New-York, Chicago, Topeka / ;
```

When elements follow a sequence, use asterisks:

```
Set t time periods /1991*2000/ ;
Set m machines /mach1*mach24/ ;
```

Transportation Example:

THE GAMS MODEL (DATA)

GAMS uses three formats for entering data:

- Lists (parameters)
- Tables
- Direct assignments

Transportation Example:
THE GAMS MODEL (Parameters)

```
Parameters

     a(i)   capacity of plant i in cases
       /     seattle       350
             san-diego     600   /

     b(j)   demand at market j in cases
       /     new-york      325
             chicago       300
             topeka        275   / ;
```

- Declare parameters and their domains a(i) and b(j)
- Values are listed between slashes / … /
- Element-value pairs must be separated by commas or entered on separate lines.

Transportation Example:
THE GAMS MODEL (Tables)

```
Table d(i,j)  distance in thousands of miles
                    new-york      chicago      topeka
    seattle            2.5          1.7          1.8
    san-diego          2.5          1.8          1.4   ;
```

- **Data can also be entered in convenient table form**
- **Declares the parameter and domain**

Transportation Example:
THE GAMS MODEL (Scalar)

```
Scalar f  freight in dollars per case per thousand miles  /90/ ;
```

- **A scalar is regarded as a parameter that has no domain.**

Transportation Example:
THE GAMS MODEL (Direct Assignment)

```
Parameter c(i,j)  transport cost in thousands of dollars per case ;

          c(i,j) = f * d(i,j) / 1000 ;
```

- When data values are to be calculated, you first declare the parameter
- then give its algebraic formulation. GAMS will automatically make the calculations.
- You can enclose the elements' names in quotes:

```
c('Seattle','New-York') = 0.40 ;
```

- The same parameter can be assigned a value more than once. Each assignment statement takes effect immediately and overrides any previous values.

Transportation Example:
THE GAMS MODEL (Variables)

```
Variables
       x(i,j)   shipment quantities in cases
       z        total transportation costs in thousands of dollars ;

Positive Variable x ;
```

- **Decision variables are expressed algebraically, with their indices specified.**
- **Variables type are: *FREE, POSITIVE, NEGATIVE, BINARY*, or *INTEGER*. The default is**
- **The objective variable (*z*, here) is to be declared without an index and should be *FREE*.**

Transportation Example:
THE GAMS MODEL (Bounds)

.lo = lower bound

.up = upper bound

```
x.up(i,j) = capacity(i,j) ;
x.lo(i,j) = 10.0 ;
x.up('seattle','new-york') = 1.2*capacity(seattle','new-york') ;
```

Transportation Example:
THE GAMS MODEL (Equations)

```
Equations
     cost          define objective function
     supply(i)     observe supply limit at plant i
     demand(j)     satisfy demand at market j ;

cost ..          z  =e=  sum((i,j), c(i,j)*x(i,j)) ;

supply(i) ..    sum(j, x(i,j))  =l=  a(i) ;

demand(j) ..    sum(i, x(i,j))  =g=  b(j) ;
```

- **Objective function and constraint equations are first declared by giving them names.**
- **Then their general algebraic formulae are described.**
- **=*e*= indicates 'equal to'**
  =*l*= indicates 'less than or equal to'
  =*g*= indicates 'greater than or equal to'

Transportation Example:
THE GAMS MODEL (Model Statements)

```
Model transport /all/ ;
```

- The model is given a unique name (here, *TRANSPORT*), and the modeler specifies which equations should be included in this particular formulation (in this case we specified ALL).
- This would be equivalent to

```
Model transport /cost, supply, demand/;
```

- This equation selection enables you to formulate different models within a single GAMS input file, based on the same or different given data.

Transportation Example:
THE GAMS MODEL (Solve Statements)

```
Solve transport using lp minimizing z ;
```

- **Tells GAMS which model to solve,**
- **Selects the solver to use (in this case an LP solver),**
- **Indicates the direction of the optimization, either *MINIMIZING* or *MAXIMIZING* ,**
- **Specifies the objective variable.**

Transportation Example:
THE GAMS MODEL (Display Statements)

```
Display x.l, z.l ;
```

- That calls for a printout of the final levels, x.l, z.l

OUTPUT FILE:

```
----      48 VARIABLE x.L  shipment quantities in cases

             new-york     chicago      topeka

seattle        50.000     300.000
san-diego     275.000                 275.000

          48 VARIABLE z.L                    =       153.675  total transportation
                                                              costs in thousands of
                                                              dollars
```

Transportation Example:
THE GAMS MODEL (Dollar Condition)

- $(condition) can be read as "***such that condition is valid***"

  if (b>1.5), then a = 2 → `a$(b>1.5)=2 ;`

- For dollar condition on the left-hand side, no assignment is made unless the logical condition is satisfied.

```
c(i,j)$(ord(i)=1) = f * d(i,j) / 1000 ;

c(i,j)$(ord(i)=card(i)-1) = f * d(i,j) / 1000 ;
```

- For dollar condition on the right hand side, an assignment is always made. If the logical condition is not satisfied, the corresponding term that the logical dollar condition is operating on evaluates to 0.
- ***if-then-else*** type of construct is implied.

```
c(i,j) = (f * d(i,j) / 1000)$(ord(i)=card(i)-1) ;
```